

4. ANALIZA STANÓW NIEUSTALONYCH

Podczas analizy stanów nieustalonych obliczana jest odpowiedź czasowa układu w zadanym przedziale $\langle 0, T \rangle$. W programie PSpice przyjęto arbitralnie, że obliczenia zaczynają się w chwili $t=0[s]$. Symulator wybiera w przedziale $\langle 0, T \rangle$ chwile czasu $\{0, t_1, t_2, \dots, t_n = T\}$, dla których równania różniczkowe zwyczajne, opisujące obwód, przybliżane są przez nieliniowe równania różnicowe. Rozwiązanie równań różnicowych dla chwil $\{0, t_1, t_2, \dots, t_n = T\}$ sprowadza się do rozwiązania ciągu równań algebraicznych (nieliniowych). Te ostatnie rozwiązuje się dokładnie tak samo jak stałoprądowe równania obwodu podczas analizy .DC. Uzyskane rozwiązanie stanowi przybliżenie dokładnego rozwiązania równań różniczkowych w chwilach $\{0, t_1, t_2, \dots, t_n = T\}$. Błąd przybliżenia w wybranej chwili czasowej nazywany jest błędem obcięcia. Algorytm wybierający chwile czasu $\{0, t_1, t_2, \dots, t_n = T\}$, a następnie całkujący równania obwodu (znajdujący przybliżone rozwiązanie równań różniczkowych) skonstruowany jest tak aby:

- ☐ Przewidywany błąd obcięcia w każdym kroku algorytmu pozostawał na akceptowanym poziomie.
- ☐ Całkowity czas obliczeń był jak najmniejszy. Poszczególne kroki czasowe $h_n = t_n - t_{n-1}$ powinny być jak największe.
- ☐ Obliczenia nie uległy destabilizacji. Objawia się to lawinowym narastaniem błędów całkowania, pojawieniem się w rozwiązaniu składowych okresowych a w rezultacie całkowitym zafałszowaniem przebiegu.

Uwzględnienie tych postulatów powoduje, że obliczone przez program PSpice przebiegi czasowe są próbkowane nierównomiernie — bardzo gęsto tam, gdzie przebieg zmienia się szybko (małe stałe czasowe) i rzadko tam, gdzie przebieg zmienia się wolno (duże stałe czasowe lub stany ustalone). Warunki początkowe panujące na elementach dynamicznych obwodu obliczane są przy założeniu, że przed chwilą $t=0$ w układzie panował stan ustalony. Można je także narzucić w sposób arbitralny używając w tym celu deklaracji .IC (ang. initial conditions — warunki początkowe).

4.1. Instrukcja analizy stanów nieustalonych

Składnia instrukcji zlecającej programowi PSpice przeprowadzenie analizy stanów nieustalonych jest następująca:

.TRAN _krok _stop [_start [_tmax]] [UIC]

Przykłady:

```
.TRAN 1NS 100NS
.TRAN 1NS 1000NS 500NS
.TRAN 10NS 100US 50US 5NS UIC
```

Analiza przeprowadzana jest w przedziale czasu $\langle 0, \text{_stop} \rangle$, natomiast wyniki wyświetlane są w przedziale $\langle \text{_start}, \text{_stop} \rangle$. Wartość parametru **_start** nie może być przy tym mniejsza niż zero. Jeśli parametr **_start** zostanie pominięty, program PSpice przyjmuje dla niego wartość zero. Wyniki wyprowadzane są za pomocą instrukcji **.PRINT** i instrukcji **.PLOT** z krokiem czasowym określonym przez parametr **_krok**. Wartość tego parametru nie może być ujemna. Parametr **_tmax** określa natomiast maksymalną wartość kroku w procesie całkowania równań układu. Zmieniając wartość tego parametru można w pewnym stopniu wpływać na proces całkowania. Użycie słowa kluczowego **UIC** powoduje, że program PSpice nie oblicza warunków początkowych na elementach dynamicznych lecz przyjmuje je tak, jak określa je deklaracja **.IC** lub tak, jak podano po słowie kluczowym **IC=** w deklaracjach elementów dynamicznych. Przy braku słowa kluczowego **UIC** warunki początkowe obliczane są przy założeniu, że wcześniej panował w układzie stan ustalony.

Deklaracja **.IC** (ang. *initial conditions* — warunki początkowe) określa warunki początkowe panujące na elementach dynamicznych w obwodzie. Jej składnia jest następująca:

.IC V(n_w1)=_war1 V(n_w2)=_war2 ...

Przykład:

```
.IC V(1)=0V V(2)=2.1V V(5)=11.69V
```

Interpretacja deklaracji przez program PSpice zmienia się w zależności od tego czy w instrukcji **.TRAN** pojawiło się słowo kluczowe **UIC**, czy też nie.

- ☐ Jeżeli w instrukcji **.TRAN** pojawiło się słowo kluczowe **UIC**, to wyszczególnione w deklaracji **.IC** wartości potencjałów węzłowych zostaną użyte do obliczenia warunków początkowych panujących na kondensatorach, cewkach, pojemnościach diod itd. Należy przy tym zwrócić uwagę, że deklaracja **.IC** ma „mniejszy priorytet” niż słowo kluczowe **IC=** pojawiające się w deklaracjach poszczególnych elementów dynamicznych. Program PSpice przed przystąpieniem do analizy stanów nieustalonych *nie* oblicza statycznego punktu pracy układu (stanu ustalonego).
- ☐ Jeżeli w instrukcji **.TRAN** nie pojawiło się słowo kluczowe **UIC**, to program PSpice oblicza warunki początkowe na elementach dynamicznych zakładając, że w układzie panuje stan ustalony i wartości potencjałów węzłowych podane w deklaracji **.IC** są

ustalone.

Nie należy mylić deklaracji .NODESET (strona 27) z deklaracją .IC. Deklaracja .NODESET jest używana podczas obliczania statycznego punktu pracy układu. Pomaga ona w sytuacji, gdy trudno jest uzyskać zbieżność obliczeń¹. W przeciwieństwie do deklaracji .IC deklaracja .NODESET nie zmienia rozwiązania.

4.1.1. Wymuszenia

Sposób deklarowania w strukturze obwodu wymuszeń przedstawiony został już na stronie 9. Nie przedstawiono tam jednak sposobu deklarowania przebiegu czasowego wartości wymuszenia. W programie PSpice przewidziano, że wymuszenie może zmieniać się w sposób skokowy, sinusoidalny, eksponentyjalny, może być sinusoidą o modulowanej częstotliwości lub przebiegiem odcinkowo–liniowym. Deklaracja przebiegu czasowego wymuszenia ma postać pola występującego na końcu linii deklaracji omówionej na stronie 9. Przedstawiono to poniżej:

VXXXXXXX n+ n- [_parametry_DC/AC] [_przebieg_czasowy]
IXXXXXXX n+ n- [_parametry_DC/AC] [_przebieg_czasowy]

W polu **_przebieg_czasowy** mogą znaleźć się deklaracje wymienione poniżej.

Impuls

PULSE(_v1 _v2 _tn _to _tt _TT)


Tablica IV Parametry przebiegu impulsowego (PULSE).

Parametr	Komentarz	Wartość domyślna	Jednostki
_v1	początkowa wartość wymuszenia	-	[V],[A]
_v2	wartość wymuszenia "po skoku"	-	[V],[A]
_tz	czas zwłoki do zbocza narastającego	0.0	[s]
_tn	czas narastania	TSTEP	[s]
_to	czas opadania	TSTEP	[s]
_długość	czas trwania impulsu	TSTOP	[s]
_okres	okres z jakim impuls jest powtarzany	TSTOP	[s]

Przykład:

IIN 3 0 PULSE(-0.1 0.1 2NS 2NS 2NS 50NS 100NS)

¹Punkt początkowy iteracji można ustalić bliżej właściwego rozwiązania tak, że algorytm Newtona-Raphsona staje się zbieżny.



Rys.37. Przebieg impulsowy (typu **PULSE**).

Tablica IV określa znaczenie poszczególnych parametrów w deklaracji. Słowo TSTEP oznacza aktualną wartość kroku czasowego z jakim całkowane są równania obwodu. Słowo TSTOP to całkowity czas symulacji. Przebieg czasowy sygnału ilustruje Rys. 37.

Tablica V Parametry przebiegu czasowego typu **SIN**.


Parametr	Komentarz	Wartość domyślna	Jednostki
_vn	wartość "niezrównoważenia"	-	[V],[A]
_va	amplituda	-	[V],[A]
_freq	częstotliwość sinusoidy	1/TSTOP	[Hz]
_tz	czas zwłoki	0.0	[s]
_tłumienie	tłumienie przebiegu	0.0	[1/s]

Sinusoida

SIN(_vn _va _freq _tz _tau)

Przykład:

VIN 3 0 DC 1V SIN(1 1 100MEG 1NS 1E10)



Rys.38. Przebieg czasowy typu **SIN**.

Tablica V podaje znaczenie poszczególnych parametrów deklaracji. Przebieg czasowy wymuszenia ilustruje Rys. 38. Jeżeli wartość parametru **_tau** zadeklarowana zostanie równa zero to przebieg czasowy jest „czystą”, nietłumioną sinusoidą.

Eksponenta


EXP(_v1 _v2 _tz1 _tau1 _tz2 _tau2)

Tablica VI Parametry przebiegu typu **EXP**.

Parametr	Komentarz	Wartość domyślna	Jednostki
_v1	wartość początkowa	-	[V],[A]
_v2	wartość "po skoku"	-	[V],[A]
_tz1	czas zwłoki do zbocza narastającego	0.0	[s]
_tau1	stała czasowa zbocza narastającego	TSTEP	[s]
_tz2	czas zwłoki do zbocza opadającego	_tz1+TSTEP	[s]
_tau2	stała czasowa zbocza opadającego	TSTEP	[s]

Przykład:

VIN 2 34 DC 1V AC 1 EXP(1 4 2NS 30NS 60NS 40NS)



Rys.39. Przebieg czasowy typu **EXP**.

Tablica VI podaje znaczenie poszczególnych parametrów deklaracji. Przebieg czasowy wymuszenia ilustruje Rys. 39. Aby opisać go analitycznie zdefiniujmy najpierw funkcję pomocniczą $e1(x, \tau)$:

$$e1(x, \tau) = \left[1 - \exp\left(-\frac{x}{\tau}\right) \right] \cdot 1(x) \quad (49)$$

Jest to funkcja równa zero dla $x < 0$ i narastająca eksponentalnie od wartości 0 do wartości 1 ze stałą czasową τ dla $x > 0$. Przebieg typu **EXP** zrealizowany w programie PSpice można teraz zapisać następująco:

$$\begin{aligned} EXP(v1, v2, tz1, \tau_1, tz2, \tau_2) = \\ = v1 + (v2 - v1) \cdot [e1(t - tz1, \tau_1) - e1(t - tz2, \tau_2)] \end{aligned} \quad (50)$$

gdzie t oznacza czas.

Sinusoida o modulowanej częstotliwości

SFFM(v_n v_a f_n i_m f_s)

Przykład:

IIN 2 3 SFFM(0 1M 20K 5 1K)

Tablica VII określa znaczenie parametrów deklaracji. Przebieg czasowy sygnału opisuje wzór:

$$SFFM(_vn, _va, _fn, _im, _fs) =$$

$$= _vn + _va \cdot \sin[2\pi \cdot _fn \cdot t + _im \cdot \sin(2\pi \cdot _fs \cdot t)] \quad (51)$$

gdzie t oznacza czas.

Tablica VII Parametry przebiegu typu **SFFM**.

Parametr	Komentarz	Wartość domyślna	Jednostki
$_vn$	niezrównoważenie	-	[V],[A]
$_va$	amplituda sygnału nośnego	-	[V],[A]
$_fn$	częstotliwość sygnału nośnego	1/TSTOP	[Hz]
$_im$	indeks modulacji	-	[-]
$_fs$	częstotliwość sygnału modulującego	1/TSTOP	[Hz]

Przebieg odcinkowo–liniowy

PWL($_t1$ $_w1$ [$_t2$ $_w2$...])

Przykład:

VCLOCK 2 3 PWL(0NS -7 10NS -7 11NS -3 17NS -3 18NS -7 50NS -7)

Para wartości ($_ti, _wi$) umieszczona jako argument po słowie kluczowym PWL oznacza, że dla chwili $t = _ti$ wymuszenie ma wartość $_wi$ (w amperach lub woltach). Par takich można umieścić w deklaracji przebiegu odcinkowo–liniowego dowolną liczbę. Wartości przebiegu dla chwil czasowych leżących między wartościami $_t1, _t2, \dots$ oblicza się za pomocą interpolacji liniowej.

4.1.2. Klucze

Elementami, które pojawiają się praktycznie tylko podczas analizy stanów nieustalonych są klucze. Program PSpice dopuszcza stosowanie dwóch typów kluczy tj. klucza sterowanego napięciem i klucza sterowanego prądem. Deklaracja klucza sterowanego napięciem w strukturze obwodu ma postać:

SXXXXXXX $n1+$ $n1-$ $n2+$ $n2-$ $_n_modelu$

Podobnie, deklaracja klucza sterowanego prądem w strukturze obwodu ma postać:

WXXXXXXX $n1+$ $n1-$ $_nazwa_SEM$ $_n_modelu$

Przykłady:

SRESET 3 4 15 8 RELAY

W12 3 7 VCI WMOD

Klucz wpięty jest między węzły, których numery określone są przez parametry **n1+** i **n1-**. W przypadku klucza sterowanego napięciem, napięcie sterujące to różnica potencjałów między węzłami obwodu o numerach **n2+** i **n2-**. W przypadku klucza sterowanego prądem, prąd sterujący płynie przez siłę elektromotoryczną o nazwie określonej przez parametr **_nazwa_SEM**. Ostatni parametr w linii deklaracji klucza to **_n_modelu**. Określa on nazwę modelu klucza. Model klucza deklarowany jest za pomocą deklaracji **.MODEL**. Deklaracja ta dla klucza sterowanego napięciem przyjmuje postać:

.MODEL _n_modelu VSWITCH [RON=_w1][ROFF=_w2][VON=_w3][VOFF=_w4]

Tablica VIII Parametry modelu klucza sterowanego napięciem.

Parametr	Komentarz	Wartość domyślna	Jednostki
RON	oporność w stanie zwarcia	1.0	[Ω]
ROFF	oporność w stanie rozwarcia	1.0MEG	[Ω]
VON	napięcie powodujące zwarcie	1.0	[V]
VOFF	napięcie powodujące rozwarcie	0.0	[V]

Dla klucza sterowanego napięciem deklaracja modelu ma postać:


.MODEL _n_modelu ISWITCH [RON=_w1][ROFF=_w2][ION=_w3][IOFF=_w4]

Tablica IX Parametry klucza sterowanego prądem.

Parametr	Komentarz	Wartość domyślna	Jednostki
RON	oporność w stanie zwarcia	1.0	[Ω]
ROFF	oporność w stanie rozwarcia	1.0MEG	[Ω]
ION	prąd powodujący zwarcie	0.01	[V]
IOFF	prąd powodujący rozwarcie	0.0	[V]

Tablica VIII oraz Tablica IX określają znaczenie poszczególnych parametrów modelu klucza.

W programie PSpice klucz realizowany jest jako element, którego oporność jest sterowana napięciem (prądem). Przykładowa zależność oporności klucza, rozumianej jako stosunek napięcia panującego na kluczu i prądu płynącego przez klucz, od napięcia sterującego przedstawiona jest na Rys. 40. Przyjęto następujące parametry klucza: RON=1[Ω], ROFF=10[Ω], VON=-1[V], VOFF=1[V]. Klucz jest zatem rodzajem elementu nieliniowego. W związku z tym stosując podczas analizy obwodu klucze należy pamiętać o następujących zasadach [26]:



Rys.40. Zależność oporności klucza od wartości napięcia sterującego. Parametry klucza: $R_{ON}=1[\Omega]$, $R_{OFF}=10[\Omega]$, $V_{ON}=-1[V]$, $V_{OFF}=1[V]$.

- ☐ Stosunek R_{OFF}/R_{ON} nie powinien być większy niż 10^{12} .
- ☐ Zakres przełączania klucza (zakres napięć sterujących od V_{ON} do V_{OFF} lub prądów sterujących od I_{ON} do I_{OFF}) nie powinien być zbyt wąski — przełączający się klucz stanowi wzmacniacz o bardzo dużym wzmocnieniu — próba rozwiązania obwodu zawierającego taki element związana jest często z poważnymi problemami numerycznymi.
- ☐ W obszarze przełączania klucza program PSpice wykonuje obliczenia z bardzo drobnym krokiem czasowym. Zbyt wiele przełączeń klucza podczas symulacji może prowadzić do długich czasów obliczeń.

4.1.3. Linia długa

Program PSpice pozwala także na modelowanie układów zawierających w swojej strukturze *bezstratną* linię długą. Deklaracja tego elementu przedstawiona została już krótko na stronie 8. Przypomnijmy ją jednak jeszcze raz:

```
TXXXXXXX n1+ n1- n2+ n2- Z0=_w1 [TD=_w2][F=_w3 [NL=_w4]]
+ [IC=_v1,_i1,_v2,_i2]
```

Przykład:

TDELAY 1 2 7 5 Z0=50 TD=1US

Zaciski wejściowe linii stanowią węzły o numerach **n1+** i **n1-**, zaciski wyjściowe to węzły o numerach **n2+** i **n2-**. Po słowie kluczowym **Z0=** należy podać impedancję falową linii wyrażoną w omach. Czas przejścia fali elektromagnetycznej od początku do końca linii może zostać podany na dwa różne sposoby:

- ☐ Po słowie kluczowym **TD=** podaje się czas (w sekundach) potrzebny na to by fala elektromagnetyczna przebyła linię.
- ☐ Po słowie kluczowym **F=** podaje się częstotliwość fali, o kształcie sinusoidy, propagującej się w linii. Po słowie kluczowym **NL=** należy podać długość linii transmisyjnej. Jednostką długości jest w tym przypadku długość fali propagującej się w linii. Jeżeli parametr NL nie zostanie podany, to zostanie przyjęte, że NL=0.25, tzn. obliczenia będą prowadzone dla linii ćwierćfalowej.

Oba sposoby podawania czasu propagacji są sobie równoważne. Należy jednak pamiętać, że każda deklaracja linii długiej *musi* określać czas propagacji fali elektromagnetycznej przez deklarowaną linię długą.

Po słowie kluczowym **IC=** można podać wartość prądu i napięcia na wejściu linii (**_v1, _i1**) oraz na wyjściu linii (**_v2, _i2**) w chwili $t=0$. Jeżeli w instrukcji .TRAN użyje się opcji UIC pozwala to, określić warunki panujące na wejściu i wyjściu linii w chwili rozpoczęcia przez program PSpice analizy stanów nieustalonych.

Program PSpice dokonując analizy .TRAN ogranicza wielkość kroku całkowania do wartości nie przekraczającej 1/2 czasu propagacji fali elektromagnetycznej przez linię. Uwzględnienie w strukturze analizowanego obwodu linii transmisyjnej, której czas propagacji jest mały w stosunku do czasów charakterystycznych dla reszty obwodu powoduje niepotrzebne powiększenie czasu zużywanego przez jednostkę centralną maszyny cyfrowej na dokonanie analizy stanów nieustalonych.

Przykład:

Obliczyć przebiegi na wejściu i wyjściu

Tablica X Parametry modelu diody dla standardowej bramki TTL.


Parametr	Wartość	Jednostki
IS	10^{-14}	[A]
RS	40	[Ω]
TT	0.1	[ns]
CJO	0.9	[pF]
VJ	1.0	[V]

linii transmisyjnej o impedancji falowej równej $100[\Omega]$ i czasie propagacji równym $10[\text{ns}]^2$. Linia sterowana jest przez bramkę TTL i obciążona jest taką samą bramką (Rys. 41). Struktura standardowej bramki TTL pokazana jest na Rys. 42. Parametry statyczne i dynamiczne modelu tranzystora należy przyjąć takie, jak w rozdziale 3, gdzie modelowano wzmacniacz operacyjny $\mu\text{A}741$ — Tablica III. Układ ten wykonywany jest w technologii zbliżonej do tej, która używana jest do wytwarzania standardowych bramek TTL. Parametry modelu diody zawiera Tablica X. Zostały one zaczerpnięte z tego samego źródła co parametry modelu tranzystora — [24]. Impuls sterujący bramkę wejściową zmienia się od $0[\text{V}]$ do $5[\text{V}]$ w czasie $5[\text{ns}]$. Czas opadania impulsu jest identyczny. Czas trwania impulsu równy jest $100[\text{ns}]$.

Dane dla programu PSpice przedstawione są poniżej. Strukturę bramki TTL (NOT) zadeklarowano w postaci podobwołu i umieszczono w osobnym zbiorze o nazwie NOT.CIR. Na zewnątrz podobwołu dostępne jest wejście bramki, wyjście oraz zacisk zasilania. Dla prostoty zbiór ten przedstawiono zaraz za zbiorem zawierającym deklarację obwodu.

```
Transmisja sygnałów TTL przez linię długą
.INC NOT.CIR
VCC 10 0 5V
VIN 1 0 DC 0 PULSE(0 5 1N 5N 5N
100N)
X1 1 2 10 NOT
X2 3 4 10 NOT
T1 2 0 3 0 Z0=100 TD=10N


.TRAN 1N 200N
.PROBE V(2) V(3)
.END
```



Rys.41. Przesyłanie sygnałów TTL przez linię transmisyjną.

```
*Zawartość zbioru NOT.CIR
.SUBCKT NOT 1 8 3
*
*      WEjście      WYjście      zasilanie
*
.MODEL TRA NPN BF=20 BR=1 IS=1.0E-14
+RB=70 RC=40 VA=50
+TF=0.1N TR=10N CCS=2.0P CJE=0.9P
+CJC=1.5P VJE=0.85 VJC=0.85 VJS=0.85
.MODEL TRB NPN BF=20 BR=0.2 IS=1.6E-14
+RB=20 RC=12 VA=50
+TF=0.1N TR=10N CCS=2.0P CJE=0.9P
+CJC=1.5P VJE=0.85 VJC=0.85 VJS=0.85
.MODEL TRC NPN BF=20 BR=0.02 IS=1.0E-14
+RB=500 RC=40 VA=50
+TF=0.1N TR=10N CCS=2.0P CJE=0.9P
+CJC=1.5P VJE=0.85 VJC=0.85 VJS=0.85
.MODEL DA D IS=1.0E-14 RS=40
+TT=0.1N CJO=0.9P VJ=1.0
```

```
R1 3 9 4K
R2 3 4 1.6K
R3 5 0 1K
R4 3 6 100
```




Rys.42. Schemat elektryczny standardowej bramki TTL.

²Przy założeniu, że prędkość rozchodzenia się fal elektromagnetycznych w linii wynosi $2.5 \cdot 10^8[\text{m/s}]$ daje to linię transmisyjną o długości $2.5[\text{m}]$.

```
Q1 2 9 1 TRC
Q2 4 2 5 TRA
Q3 6 4 7 TRA
Q4 8 5 0 TRB
```

```
D1 7 8 DA
D2 0 1 DA
```

```
.ENDS NOT
```




Rys.43. Napięcie na wejściu V(2) i na wyjściu V(3) linii długiej, przez którą transmitowane są sygnały TTL.

Linia długa łącząca bramki modelowana jest w taki sposób, że jeden z przewodów linii jest idealnie uziemiony. Pomija się w tym przypadku składowe fali elektromagnetycznej rozchodzące się pomiędzy tym przewodem, a otoczeniem układu.

Analiza przeprowadzana jest w przedziale czasu od 0[ns] do 200[ns]. Ponieważ nie używamy ani instrukcji .PRINT, ani instrukcji .PLOT pierwszy parametr w instrukcji analizy stanów nieustalonych, oznaczający krok czasowy wydruku, jest nieistotny (jeżeli tylko jest mniejszy niż całkowity czas symulacji).

Uzyskane wyniki przedstawiono na Rys. 43. Znajdujemy tam przebieg napięcia na wejściu linii długiej V(2) oraz przebieg napięcia na wyjściu V(3). Widać, że są one nie tylko przesunięte względem siebie, ale także różnią się dość istotnie kształtem. Dla porównania na Rys. 44 przedstawiono przebieg napięcia w układzie, w którym połączenie między bramkami



Rys.44. Napięcie na wyjściu bramki TTL sterującej taką samą bramką przez połączenie krótkie.

jest połączeniem krótkim. W tym przypadku przebieg jest znacznie bardziej „gładki”.

4.1.4. Sterowanie procesem całkowania równań układu

Wróćmy do przykładu generatora Collpits-a (Rys. 45), który był rozważany w rozdziale 3 na stronie 58. Spróbujmy przeprowadzić analizę stanów nieustalonych w tym układzie. W tym celu należy przygotować następujący zestaw danych dla programu PSpice:

```
ANALIZA GENERATORA - stan nieustalony
.MODEL BC241 NPN(IS=67.34F XTI=3
+EG=1.11 VAF=100 BF=116.1 NE=3.779
+ISE=32.7N IKF=33.53M XTB=1.5
BR=1.926M NC=2 ISC=0 IKR=0 RC=1.5
+CJC=4.929P VJC=.75 MJC=.3333 FC=.5
+CJE=3.316P VJE=.75 MJE=.3333
+TR=7.77U TF=107.3P ITF=.4 +VTF=10
XTF=2 RB=10)
```

```
VCC 1 0 9V
```

```
R1 1 2 145K
```

```
R2 2 0 76K
```

```
R3 1 3 1.2K
```


```
R4 4 0 300
```

```
Q1 3 2 4 BC241
```

```
C1 6 0 10N
```

```
C2 2 0 3N
```


```
C3 3 6 100N
```



Rys.45. Generator Collpits-a

L1 6 2 10M

```
.TRAN 1U 0.3M UIC
.PROBE V(3)
.END
```




Rys.46. Drgania w układzie generatora.

Słowo kluczowe UIC w instrukcji .TRAN powoduje, że program PSpice nie oblicza stanu ustalonego w układzie przed przystąpieniem do całkowania równań czasowych. W chwili $t=0$ przyjmuje się, że napięcie na każdym z kondensatorów jest równe zero oraz że prąd płynący przez cewkę L jest równy zero. W rezultacie analizowany jest stan nieustalony pojawiający się w układzie w chwili włączenia zasilania. Na Rys. 46 przedstawiony jest obliczony przebieg napięcia wyjściowego. Jest on bardzo „poszarpany”. Szczyty przebiegu są ostre lub ścięte. Przebieg na Rys. 46 jest bowiem zbyt rzadko próbkowany aby mógł być prawidłowo odtworzony metodą interpolacji liniowej stosowaną przez program graficzny PROBE. Konieczne jest ograniczenie maksymalnej wartości kroku czasowego w procesie całkowania równań obwodu. Jak pamiętamy (strona 72) można to zrobić przez podanie odpowiedniego parametru w instrukcji .TRAN. W naszym przypadku instrukcja ta może wyglądać następująco:

```
.TRAN 1USEK 0.3MSEK 0MSEK 1USEK UIC
```

Wpływ zmiany w instrukcji .TRAN na obliczenia napięcia wyjściowego ilustruje Rys. 47.



Rys.47. Przebieg napięcia w układzie generatora. Mały błąd obciążenia.

Przebieg napięcia wyświetlany przez program PROBE jest teraz „gładki”. Częstotliwość drgań odczytana z wykresu (trzy ostatnie okresy) wynosi $\approx 37.5[\text{kHz}]$. Różnica między tą częstotliwością, a częstotliwością obliczoną w rozdziale 3 nie przekracza 2.5%. należy uznać, że rozbieżność obliczeń jest niewielka ponieważ:

- ☐ W obliczeniach prowadzonych w rozdziale 3 nie uwzględniano żadnych zjawisk nieliniowych.
- ☐ Symulacja stanu nieustalonego odbywa się ze skończoną dokładnością.
- ☐ Przebieg z Rys. 47 nie przedstawia jeszcze stanu ustalonego.

Na wartość kroku całkowania można także wpływać pośrednio przez zmniejszenie dopuszczalnej wartości błędu obciążenia. Służy do tego bezwymiarowy parametr **TRTOL** występujący w instrukcji .OPTIONS — wartość błędu obciążenia jest proporcjonalna do wartości TRTOL. Wartość domyślna parametru TRTOL wynosi 7. Można ją jednak zmienić. W przypadku rozważanego generatora należy dane wejściowe uzupełnić o następującą linię:

.OPTIONS TRTOL=0.5


powodując czternastokrotne zmniejszenie błędu obciążenia.

4.2. Układy niestacjonarne


Dzięki zastosowaniu specjalnej techniki, program PSpice pozwala modelować nie tylko klasyczne obwody elektryczne, dla których parametry poszczególnych elementów pozostają stałe, ale także obwody których elementy mają parametry zmienne w czasie. W najprostszym przypadku równania opisujące taki obwód są równaniami różniczkowymi liniowymi, których współczynniki zależne są od czasu. Równania takie nazywamy niestacjonarnymi, stąd nazwa układów — niestacjonarne.

4.2.1. Idea realizacji elementów o zmiennych w czasie parametrach [10]


Dwójnik, który realizuje zmienną w czasie przewodność składa się z opornika R o stałej wartości połączonego szeregowo ze sterowaną siłą elektromotoryczną (Rys.48. A). Wielkości sterujące to:




A) $G_{var} = 1/R \cdot \{1 - w[us(t)]\}$



B) $R_{var} = R \cdot \{1 + w[us(t)]\}$



C) $C_{var} = C \cdot \{1 - w[us(t)]\}$



D) $L_{var} = L \cdot \{1 - w[us(t)]\}$

Rys.48. Elementy zależne od czasu — idea realizacji.

☐ Napięcie na dwójniku $u(t)$.

☐ Napięcie sterujące $u_s(t)$.

Niech wartość sterowanej siły elektromotorycznej e_1 wyraża się wzorem:

$$e_1(t) = u(t) \cdot w[u_s(t)] \quad (52)$$

Funkcja $w(x)$ może być dowolnym wielomianem. Ponieważ każdą funkcję ciągłą można aproksymować z dowolną dokładnością wielomianem nie jest to istotne ograniczenie. Równanie łączące prąd $i(t)$ płynący przez nasz dwójnik z napięciem $u(t)$ panującym między jego zaciskami można teraz zapisać następująco:

$$R \cdot i(t) = u(t) - u(t) \cdot w[u_s(t)] \quad (53)$$

Stąd wartość przewodności $G_{var}=i(t)/u(t)$ dwójnika wyraża się wzorem:

$$G = \frac{1}{R} \times \{1 - w[u_s(t)]\} \quad (54)$$

Przewodność dwójnika uzależniona jest od czasu przez napięcie sterujące $u_s(t)$. Wartość napięcia sterującego można wymuszać za pomocą niezależnej siły elektromotorycznej.

Chcąc otrzymać oporność zależną od czasu, w rozważanym powyżej układzie, siłę elektromotoryczną e_1 należy uzależnić od napięcia panującego na stałej oporności R (Rys.48. B). Równanie łączące prąd płynący przez dwójnik z napięciem panującym na nim przyjmuje następującą postać:

$$R \cdot i(t) = u(t) - [R \cdot i(t)] \cdot w[u_s(t)] \quad (55)$$

Oporność dwójnika $R_{var}=u(t)/i(t)$ wyraża się zatem wzorem:

$$R_{var} = R \cdot \{1 + w[u_s(t)]\} \quad (56)$$

Dwójnik realizujący zależną od czasu pojemność jest bardzo podobny do dwójnika realizującego zależną od czasu przewodność. Stałą oporność R zastąpiono przez stałą pojemność C (Rys.48. C). Ładunek $q(t)$ zgromadzony przez pojemność związany jest z napięciem na dwójniku w sposób następujący:

$$u(t) = \frac{q(t)}{C} + u(t) \cdot w[u_s(t)] \quad (57)$$

Pojemność $C_{var}=q(t)/u(t)$ dwójnika dana jest wzorem:

$$C_{var} = C \cdot \{1 - w[u_s(t)]\} \quad (58)$$

Sposób realizacji indukcyjności zależnej od czasu przedstawia Rys.48. D. Stała co do wartości indukcyjność L połączona jest równolegle z siłą prądomotoryczną i_1 sterowaną

prądem $i(t)$ płynącym przez dwójnik i wartością napięcia sterującego $u_s(t)$. Jeżeli siła prądomotoryczna i_1 wyraża się następującym wzorem:

$$i_1 = i(t) \cdot w[u_s(t)] \quad (59)$$

to strumień magnetyczny $\phi(t)$ związany z indukcyjnością L jest równy:

$$\phi(t) = L \{ i(t) - i(t) \cdot w[u_s(t)] \} \quad (60)$$

Stąd indukcyjność dwójnika $L_{var} = \phi(t)/i(t)$ wyraża się wzorem:

$$L_{var} = L \{ 1 - w[u_s(t)] \} \quad (61)$$

Źródła sterowane występujące w strukturze każdego z elementów mogą być uzależnione od napięcia dostarczanego przez niezależną siłę elektromotoryczną. W tym przypadku opisane struktury modelują przewodność, oporność, pojemność i indukcyjność zmienne w czasie. Możliwe jest też uzależnienie wspomnianych źródeł sterowanych od dowolnego napięcia w obwodzie. w tym przypadku mamy do czynienia z sterowaną przewodnością, opornością, pojemnością i indukcyjnością.

4.2.2. Nieliniowe źródła sterowane

W opisanych strukturach elementów zależnych od czasu występują źródła napięcia (prądu) sterowane napięciem. Funkcja opisująca zależność między wielkością sterującą i sterowaną jest wielomianem. Ogólnie rzecz biorąc program PSpice dopuszcza aby źródło napięcia (prądu) sterowane było przez jedno lub więcej napięć (prądów). Wartość wymuszenia związana jest z wielkościami sterującymi poprzez wielomian dowolnego stopnia i rzędu³. Wielomian taki określa się poprzez podanie rzędu oraz kolejnych współczynników liczbowych $P_0, P_1, P_2, \dots, P_n$. Znaczenie tych współczynników zmienia się wraz ze zmianą rzędu wielomianu.

W przypadku wielomianu jednej zmiennej $w(x_1)$ wartość wielomianu obliczana jest zgodnie z następującym wzorem:

$$w(x_1) = P_0 + P_1 \cdot x_1 + P_2 \cdot x_1^2 + P_3 \cdot x_1^3 + P_4 \cdot x_1^4 + P_5 \cdot x_1^5 + \dots \quad (62)$$

W przypadku wielomianu dwóch zmiennych $w(x_1, x_2)$ jego wartość obliczana jest zgodnie z następującym wzorem:

³Przez rząd wielomianu rozumie się tutaj liczbę zmiennych niezależnych występujących w tym wielomianie.

$$w(x_1, x_2) = P_0 + P_1 \cdot x_1 + P_2 \cdot x_2 + P_3 \cdot x_1^2 + P_4 \cdot x_1 \cdot x_2 + P_5 \cdot x_2^2 + \\ + P_6 \cdot x_1^3 + P_7 \cdot x_1^2 \cdot x_2 + P_8 \cdot x_1 \cdot x_2^2 + P_9 \cdot x_2^3 + \dots \quad (63)$$

W przypadku wielomianu trzech zmiennych $w(x_1, x_2, x_3)$ wartość wielomianu obliczana jest zgodnie ze wzorem:

$$w(x_1, x_2, x_3) = P_0 + P_1 \cdot x_1 + P_2 \cdot x_2 + P_3 \cdot x_3 + P_4 \cdot x_1^2 + P_5 \cdot x_1 \cdot x_2 + P_6 \cdot x_1 \cdot x_3 + \\ + P_7 \cdot x_2^2 + P_8 \cdot x_2 \cdot x_3 + P_9 \cdot x_3^2 + P_{10} \cdot x_1^3 + P_{11} \cdot x_1^2 \cdot x_2 + P_{12} \cdot x_1^2 \cdot x_3 + P_{13} \cdot x_1 \cdot x_2^2 + \\ + P_{14} \cdot x_1 \cdot x_2 \cdot x_3 + P_{15} \cdot x_1 \cdot x_3^2 + P_{16} \cdot x_2^3 + P_{17} \cdot x_2^2 \cdot x_3 + P_{18} \cdot x_2 \cdot x_3^2 + P_{19} \cdot x_3^3 + P_{20} \cdot x_1^4 + \dots \quad (64)$$

Współczynniki wielomianów wyższych rzędów oznaczane są w sposób analogiczny do tego, który zilustrowano dla wielomianu rzędu od 1 do 3.

Deklaracja nieliniowego źródła napięcia (prądu) sterowanego napięciem przyjmuje postać:

EXXXXXXXX n+ n- POLY(_rz) nc1+ nc1- ... P0 [P1...] [IC=...]
GXXXXXXX n+ n- POLY(_rz) nc1+ nc1- ... P0 [P1...] [IC=...]

Przykłady:

ER12 17 0 POLY(3) 13 0 15 0 17 0 0 0 1 1 1 IC=1.5,0.17,32
 GCRT 2 13 POLY(2) 3 5 1 2 0 1M 17M 3.5U IC=2.5,1.3

Słowo kluczowe **POLY** identyfikuje nieliniowe sprzężenie między wymuszeniem, a wartościami napięć sterujących. Parametr **_rz** określa rząd wielomianu — liczbę napięć sterujących. Parametry **P0, P1, P2, ...** to współczynniki wielomianu, których znaczenie opisano wyżej. Natomiast po słowie kluczowym **IC=** można podać wartości początkowe napięć sterujących. Słowo kluczowe **IC=** zostało wprowadzone do deklaracji nieliniowego źródła sterowanego dla ułatwienia (przyspieszenia) procesu obliczania statycznego punktu pracy układu. Wartości początkowe wielkości sterujących służą programowi PSpice do obliczenia wstępnych wartości potencjałów węzłowych — podczas tego procesu nieliniowe źródła sterowane są w istocie traktowane jak zwykłe źródła niezależne. Obliczony wstępnie statyczny punkt pracy układu stanowi punkt wyjścia do obliczenia właściwego punktu pracy układu — dla nieliniowych źródeł sterowanych przywracane jest sprzężenie między wielkościami sterowanymi a sterującymi. W przypadku, gdy słowo kluczowe **IC=** nie pojawi się przyjmuje się wstępnie, że wartości napięć sterujących wynoszą zero.

Deklaracje nieliniowych źródeł sterowanych prądem mają postać analogiczną do omówionych powyżej źródeł sterowanych napięciem:

HXXXXXXX n+ n- POLY(_rz) _v1 _v2 ... _v_rz P0 [P1 ...] [IC= ...]
FXXXXXXXX n+ n- POLY(_rz) _v1 _v2 ... _v_rz P0 [P1 ...] [IC= ...]

Przykłady:


FMIX 12 20 POLY(1) VSENS 1MA 1.3M
 HAC 12 30 POLY(2) V1 V2 0 0 0 0 1 IC=0.5 1.3

Pierwszy przykład ilustruje deklarację źródła prądu sterowanego prądem płynącym przez


```

MOSTEK O ZMIENNEJ W CZASIE PO-
JEMNOŚCI
.LIB ELEMENTY.LIB
.OPTIONS TRTOL=1.0
R1 1 2 290K
R2 1 3 290K
Xvar 2 4 20 40 POJEMNOSC
VSTER 20 40 PULSE(0.3M 0.4M 0.0
500U)
C2 4 3 350P
R5 4 0 35K
C3 5 2 1U
C4 3 6 1U
R3 5 0 1MEG
R4 6 0 1MEG
VS 1 0 SIN(0 10 2KHZ)
.TRAN 1U 500U
.PROBE
.END

```



Rys.53. Mostek impedancyjny.

Na wykresie ilustrującym wyniki obliczeń (Rys. 54) umieszczono obok napięcia na przekątnej mostka w przypadku, gdy kondensator Cvar zmienia swą wartość od 300pF do 400pF, to samo napięcie w przypadku, gdy kondensator Cvar ma stałą wartość 300pF oraz 400pF.

Deklaracja struktury kondensatora o zmiennej w czasie wartości pojemności umieszczona została w zbiorze bibliotecznym o nazwie ELEMENTY.LIB. Po przeczytaniu deklaracji całego obwodu program PSpice otwiera zbiór ELEMENTY.LIB, wymieniony w instrukcji .LIB, i odszukuje deklarację struktury o nazwie „pojemnosc”, której nie znalazł w zbiorze z danymi wejściowymi.


Instrukcja dołączania bibliotek .LIB nakazuje programowi PSpice poszukiwanie deklaracji i modeli nie zamieszczonych bezpośrednio w zbiorze z danymi w zbiorze bibliotecznym. Składnia tej instrukcji jest następująca:

.LIB _nazwa_zbioru

Przykład:

.LIB c:\spice\bib\diody.lib

Parametr **_nazwa_zbioru** określa nazwę zbioru bibliotecznego, który będzie przeglądany przez program PSpice. Nazwa zbioru może być poprzedzona ścieżką i specyfikacją napędu dyskowego. Instrukcja .LIB może być zagnieżdżana tzn. zbiór dołączany za pomocą instrukcji .LIB może zawierać instrukcję .LIB. Liczba zagnieżdżeń jest nieograniczona. W przypadku, gdy parametr **_nazwa_zbioru** jest ciągiem pustym, przeglądana jest biblioteka o nazwie NOM.LIB, o której zakłada się, że jest w katalogu domyślnym. Wygodnie jest zatem nazwy



Rys.54. Napięcie na przekątnej mostka impedancyjnego.

@1 $C_{var}=300pF$; @2 $C_{var}=400pF$; @3 $C_{var}=300pF+100pF*TIME/0.5ms$.

wszystkich posiadanych bibliotek umieścić w zbiorze NOM.LIB⁴. Jest to uzasadnione tym bardziej, że w odróżnieniu od instrukcji .INC (strona 67) program PSpice *przetwarza tylko te deklaracje i modele, które są potrzebne do pełnego zdefiniowania obwodu*.

Zwróćmy uwagę na okoliczność, że w przypadku każdego z elementów zawartych w bibliotece ELEMENTY.LIB napięciem sterującym może być dowolne napięcie występujące w obwodzie. Biblioteka jest zatem czymś więcej niż tylko biblioteką elementów zależnych od czasu. Jest to biblioteka elementów, których wartość może być sterowana dowolnym napięciem w obwodzie. Wartość przewodności, oporności, pojemności, indukcyjności jest wprost proporcjonalna do wartości napięcia sterującego.

Ze względu na wymaganie, które narzuca program PSpice, aby każdy węzeł analizowanego obwodu miał stałoprądowe połączenie z węzłem masy zdecydowano, że:

- Oporność różnicowa każdego z elementów sterowanych, widziana z zacisków sterujących, będzie wynosić 100[MΩ].
- Oporność widziana między węzłem masy i każdym z zacisków sterujących będzie wynosić ok. 1000[MΩ].

⁴Jako parametry instrukcji .LIB.

Jeżeli oporności te zakłócają zbytnio pracę układu, natomiast węzły sterujące połączone są z masą poza elementem sterowanym oporności te można usunąć.

Przedstawiona w podrozdziale 4.2.1 idea stanowić może podstawę rozbudowy biblioteki ELEMENTY.LIB o następujące elementy:

- ☐ Elementy, których wartość związana jest z napięciem sterującym dowolnym wielomianem. My zrealizowaliśmy elementy zależne liniowo od napięcia sterującego.
- ☐ Elementy, których wartość sterowana jest przez dowolny prąd płynący w obwodzie.

4.3. Analiza zniekształceń nieliniowych

Przebiegi uzyskane podczas analizy w dziedzinie czasu (analizy stanów nieustalonych) mogą zostać poddane analizie widmowej. Najczęściej jej celem jest określenie współczynnika zniekształceń nieliniowych przebiegu. Program PSpice łączy obie operacje tj. analizę widmową i analizę zniekształceń nieliniowych w jednej instrukcji .FOUR (ang. fourier analysis — analiza fourierowska). Składnia tej instrukcji jest następująca:

.FOUR _częstotliwość _wy1 [_wy2 ...]

Przykład:

.FOUR 1k V(2,5) I(VIN)


Instrukcja .FOUR musi być poprzedzona instrukcją analizy stanów nieustalonych, podczas której obliczone zostaną przebiegi czasowe wielkości określonych przez parametry **_wy1** **_wy2 ...** Program PSpice oblicza składowe harmoniczne przebiegów w przedziale czasowym **<_stop-1/_częstotliwość, _stop>**. Parametr **_stop** oznacza chwilę czasową, dla której zakończono obliczanie stanów nieustalonych. Parametr **_częstotliwość** to częstotliwość (wyrażona w hertzach) składowej podstawowej przebiegu. Obliczana jest wartość (amplituda i faza) pierwszych dziesięciu składowych harmonicznych analizowanego przebiegu oraz współczynnik zniekształceń nieliniowych. W praktyce wartość parametru **_stop**, w instrukcji .TRAN, dobiera się tak aby do chwili **_stop-1/_częstotliwość** zanikły w obwodzie wszystkie stany nieustalone.

Przykład:

Obliczyć zależność współczynnika zniekształceń nieliniowych od napięcia sterującego dla układu wzmacniacza mocy zbudowanego w oparciu o układ scalony UL1461L. Założyć, że układ sterowany jest ze źródła o oporności 1[kΩ] sygnałem o częstotliwości 1[kHz].

Schemat aplikacyjny układu UL1461L pokazano na Rys. 56. Schemat elektryczny scalonego wzmacniacza mocy UL1461L pokazano na Rys. 55. Oba zaczerpnięto z [21]. Tablica III i Tablica X zawierają wartości parametrów modelu tranzystora i modelu diody przyjęte do obliczeń. Strukturę układu scalonego zadeklarowano w postaci podobwołu o nazwie UL1461L. Podobwód ten umieszczony został w osobnym zbiorze (o nazwie UL1461L.CIR). W ten sposób może być wykorzystany do innych obliczeń.


```
.SUBCKT UL1461L 1 2 3 4 5 6 7 8 9 10 11 12 13
*
*
*   numery wyprowadzeń układu scalonego
*
R1      10      20
7.5K
R3      20      11
25K
R2      20      21
1K
R4      24      25
15K
R5      10      50
30K
R6      25      13
1K
R7      26      12
380
R8      26      27
5K
R9      27      9
12.5K
R10     27      28
3.75K
R11     10      30
750
R12     30      9
250
R13     33      1
260
R14     10      6
875
R15     8       2
8.5K
R16     2       13      3K
R17     6       2       3K
R18     35      13      5.1K
R19     6       34      3K
R20     39      3       310
R21     6       36      270
R22     4       13      1K
R23     6       43      100
R24     38      13      550
R25     6       7       240
R26     40      13      550
R27     5       39      10K
R28     44      5       550
```



Rys.55. Schemat elektryczny układu scalonego 1461L.

```
.MODEL TM NPN BF=80 BR=1 IS=1E-14 RB=100 VA=50 TF=0.3N TR=6N CJS=2.0P CJE=3.0P CJC=2.0P
.MODEL TM1 PNP BF=10 BR=1 IS=1E-14 RB=20 VA=50 TF=0.3N TR=20N CJS=0.0P CJE=6.0P CJC=4.0P
Q1      10      11      24      TM
Q2      50      24      25      TM
Q3      10      26      25      TM
Q4      30      50      32      TM
Q5      30      32      31      TM
Q6      9       31      33      TM
Q7      6       8       35      TM
Q8      34      39      35      TM
Q9      37      34      36      TM1
Q10     36      37      4       TM
Q11     43      4       38      TM
Q12     7       38      40      TM
Q13     41      40      13      TM 10
Q14     6       7       44      TM
Q15     6       44      5       TM 10

.MODEL DIO D IS=1E-14 TT=0.1N CJO=0.9P VJ=1.0
D1      21      22      DIO
D2      22      13      DIO
D3      28      13      DIO
D4      7       42      DIO
D5      42      41      DIO
D6      5       41      DIO
.ENDS
```

Linie, w których zadeklarowano tranzystory Q13 i Q15 zawierają dodatkowy parametr. W dokumentacji programu PSpice nosi on nazwę AREA. Oznacza on o ile większe pole powierzchni krzemu zajmują tranzystory Q13 i Q15 w stosunku do reszty tranzystorów opisywanych tym samym modelem. Założyliśmy, że tranzystory Q13 i Q15 zajmują 10 krotnie większą powierzchnię, niż pozostałe tranzystory. Uzasadnieniem jest fakt, że tranzystory Q13 i Q15 połączone są bezpośrednio z wyjściem układu scalonego. Płynie przez nie znacznie większy prąd niż przez pozostałe elementy. Wymaga to skuteczniejszego odprowadzania ciepła. W przypadku układu scalonego uzyskuje się to przez powiększenie powierzchni zajmowanej przez tranzystor. Odpowiednio do wartości parametru AREA zmieniana jest wartość parametrów⁵ zależnych od powierzchni. W ten sposób jedna deklaracja modelu tranzystora wystarcza dla opisu zachowania wszystkich tranzystorów, które powstały w jednym procesie technologicznym.


Model układu scalonego dołączany jest do deklaracji struktury wzmacniacza mocy za pomocą instrukcji .INC. Równie dobrze można byłoby użyć instrukcji .LIB — instrukcji dołączania biblioteki.

```
WZMACNIACZ MOCY NA UL1461L -
2.7
.OPTIONS TRTOL=1 ITL5=10000
.INC UL1461L.CIR
```

```
VWE      30      0      AC 0.5M
SIN(0 2.7M 1K)
RWE      20      30      1K
VCC      6      0      DC
13.2V
```

```
C1      11      20      10U
C2      12      0      10U
C3      10      0      220U
C4      8      0      15N
C5      21      9      10U
C6      7      6      4.7N
C7      6      0      470U
C8      5      22      470U
C9      5      4      560P
C10     4      0      1N
C11     3      0      47U
C12     2      0      22U
```

```
R1      21      8      1K
ROBC    22      0      4
```




Rys.56. Wzmacniacz mocy zbudowany w oparciu o układ scalony UL1461L.

```
XPA      0 2 3 4 5 6 7 8 9 10 11 +12 0 UL1461L
*.AC DEC 10 1 30K
.TRAN 0.1M 20M 10U
.FOUR 1K V(22)
.PROBE V(22)
.END
```

⁵Wśród zmienianych parametrów są: prądy nasycenia złącz, rezystancja rozproszona obszaru bazy, pojemności złączowe i inne.

Jako pierwszą przeprowadzono małosygnałową analizę zmiennoprądową wzmacniacza (odpowiednia instrukcja „zamaskowana” jest znakiem początku komentarza „ * ”). W jej wyniku okazało się, że pasmo przenoszenia analizowanego układu rozciąga się od około 100[Hz] do 10[kHz]. Wzmocnienie układu wynosi zaś $\approx 2[\text{V/mV}]$ ($\approx 66\text{dB}$). Biorąc pod uwagę, że napięcie zasilania wynosi 13.2[V] należy się spodziewać, że przy amplitudzie sygnału wyjściowego $\approx (13.2 - 2 \cdot 0.7)/2 = 5.9\text{[V]}$ wystąpią już istotne zniekształcenia nieliniowe sygnału wyjściowego. Odpowiada to sygnałowi $\approx 2.8\text{[mV]}$ na wejściu układu. Przewidywania te potwierdza wynik analizy Fouriera przebiegu wyjściowego, dla różnych wartości amplitudy napięcia sterującego — Rys. 57. W przedziale wartości amplitudy napięcia sterującego od 2.2[mV] do 2.3[mV] wielkość współczynnika zniekształceń nieliniowych przekracza wartość 1.0%.

Wyniki analizy Fouriera umieszczane są przez program PSpice w zbiorze wyjściowym. Przykładowy fragment danych wyjściowych pokazany jest poniżej.



Rys.57. Współczynnik zniekształceń nieliniowych układu wzmacniacza mocy w funkcji napięcia sterującego.

FOURIER COMPONENTS OF TRANSIENT RESPONSE V(22)


DC COMPONENT = 1.553569E-02

HARMONIC NO	FREQUENCY (HZ)	FOURIER COMPONENT	NORMALIZED COMPONENT	PHASE (DEG)	NORMALIZED PHASE (DEG)
1	1.000E+03	5.617E+00	1.000E+00	1.763E+00	0.000E+00
2	2.000E+03	1.819E-01	3.239E-02	8.613E+01	8.437E+01
3	3.000E+03	2.252E-01	4.009E-02	-7.215E+00	-8.978E+00
4	4.000E+03	1.105E-01	1.967E-02	-9.849E+01	-1.003E+02
5	5.000E+03	1.187E-01	2.114E-02	1.673E+02	1.655E+02
6	6.000E+03	1.075E-02	1.915E-03	6.725E+01	6.548E+01
7	7.000E+03	2.987E-02	5.317E-03	-2.291E+01	-2.468E+01
8	8.000E+03	2.605E-02	4.637E-03	6.584E+01	6.408E+01
9	9.000E+03	1.462E-03	2.603E-04	-1.827E+01	-2.003E+01

TOTAL HARMONIC DISTORTION = 5.953000E+00 PERCENT

Wyniki zawierają wartość składowej stałej przebiegu oraz wartość dziewięciu kolejnych składowych harmoniczných. Dla każdej składowej podane są kolejno: wartość częstotliwości, wartość amplitudy, wartość amplitudy odniesiona do pierwszej harmonicznej, przesunięcie fazowe oraz przesunięcie fazowe odniesione do pierwszej harmonicznej.

Uzyskane rezultaty w obszarze małych wartości współczynnika zniekształceń nieliniowych należy przyjmować z pewną ostrożnością. Do rzeczywistych zniekształceń nieliniowych




Rys.58. Przebieg napięcia wyjściowego w układzie wzmacniacza mocy. Amplituda napięcia sterującego wynosi 2.7[mV].

dołączają błędy wynikające z aproksymacji liniowej przebiegu między punktami, w których wartość przebiegu została obliczona w procesie całkowania równań układu. Z tego względu w instrukcji .TRAN ograniczona została maksymalna wartość kroku całkowania do 10[μs] — dane dla programu PSpice na stronie 96. Stanowi to 1.0% okresu składowej podstawowej i można się spodziewać, że jeżeli wartość rzeczywistego współczynnika zniekształceń nieliniowych jest powyżej 1.0%, to program PSpice obliczy go poprawnie.

Dla napięcia sterującego równego 2.7[mV] napięcie wyjściowe w analizowanym wzmacniaczu mocy jest już poważnie zniekształcone. Przebieg napięcia wyjściowego przedstawia Rys. 58. Obliczony przez program PSpice współczynnik zniekształceń nieliniowych wynosi prawie 6[%]. Za pomocą programu PROBE można obliczyć *ciągłe* widmo przebiegu z Rys. 58. Służy temu opcja **Fourier** figurująca w menu programu. Wynik przedstawia Rys. 59. Dokonując analizy widmowej za pomocą programu PROBE należy mieć na uwadze, że obliczana jest *dyskretna* transformata Fouriera (przy pomocy algorytmu FFT). Liczba próbek sygnału pobieranych do przetwarzania jest ograniczona. Oznacza to, że:

- Przebieg w dziedzinie czasu będzie zbyt rzadko próbkowany (szerokie okno czasowe) to dojdzie do zniekształcenia widma przebiegu w wyniku zjawiska aliasingu (twierdzenie Kotelnikowa–Shanona).



Rys.59. Widmo ciągłe przebiegu z poprzedniego rysunku obliczone za pomocą programu PROBE.

- ☐ Jeżeli przebieg w dziedzinie czasu będzie zbyt często próbkowany to także dojdzie do zniekształcenia widma w wyniku zbytniego zawężenia okna czasowego.

4.4. Zastosowanie programu PSpice do problemów nieelektrycznych

Program PSpice może być traktowany jak maszyna analogowa. Załóżmy, że interesuje nas rozwiązanie pewnego problemu, który da się sformułować w postaci równań różniczkowych zwyczajnych. Jeżeli istnieje obwód elektryczny, opisywany takimi samymi równaniami tzn. istnieje analog elektryczny, to rozwiązując za pomocą programu PSpice równania tego obwodu rozwiązujemy interesujący nas problem. Poniżej przedstawiono dwa przykłady zastosowania programu PSpice do rozwiązywania problemów, które formalnie nie mają nic wspólnego z obwodami elektrycznymi.

4.4.1. Atraktor Lorentza [18]

W ostatnich latach wśród naukowców–przyrodznawców tzn. fizyków, chemików, biologów bardzo duże zainteresowanie wzbudziły wyniki badań nad rozwiązaniami pewnych nieliniowych równań różniczkowych. Okazało się, że istnieje szansa aby procesy dotychczas uznawane za chaotyczne jak np. turbulentny przepływ cieczy, zjawiska atmosferyczne czy

pewne reakcje chemiczne opisać w sposób deterministyczny — za pomocą równań różniczkowych. Dotychczas badano zjawiska opisując je za pomocą równań liniowych. Ich rozwiązania cechuje duża regularność. Brak jest miejsca na chaos i przypadkowość. Stąd przeciwstawiano sobie opis deterministyczny — za pomocą równań (różniczkowych) — oraz opis probabilistyczny — za pomocą prawdopodobieństw. Uwzględnienie w opisie deterministycznym zjawisk nieliniowych dało nieoczekiwane rezultaty — rozwiązanie równań różniczkowych może wyglądać zupełnie chaotycznie. Przykładem jest następujący układ równań różniczkowych:

$$\begin{cases} \frac{dx_1}{dt} = 10 \cdot (x_2 - x_1) \\ \frac{dx_2}{dt} = x_1 \cdot (28 - x_3) - x_2 \\ \frac{dx_3}{dt} = x_1 \cdot x_2 - \frac{8}{3} \cdot x_3 \end{cases} \quad (65)$$

Aby go rozwiązać za pomocą programu PSpice, należy zbudować analog elektryczny — obwód opisywany takimi samymi równaniami różniczkowymi. Układ ten przedstawiony jest na Rys. 60. Składają się nań trzy sekcje. Każda sekcja odpowiada jednemu równaniu w układzie równań (65). Zmienne x_1 , x_2 , x_3 reprezentowane są odpowiednio przez potencjały węzłowe V(1), V(2), V(3). Dane dla programu PSpice przedstawione są poniżej.


```
DZIWNY ATRAKTOR
.OPTIONS TRTOL=.1 ITL5=15000

*PIERWSZA SEKCJA
C1 1 0 1 IC=0
G1 0 1 2 0 10
R1 1 0 0.1

*DRUGA SEKCJA
C2 2 0 1 IC=10
G2 0 2 POLY(2) 1 0 3 0 0 28
0 0 -1
R2 2 0 1

*TRZECIA SEKCJA
C3 3 0 1 IC=27
R3 3 0 0.375
G3 0 3 POLY(2) 1 0 2 0 0 0 0
0 1


.TRAN 0.01 40 UIC
.PROBE V(1) V(2) V(3)
.END
```



Rys.60. Analog elektryczny atraktora Lorentza.

Całkowanie równań układu chcemy rozpocząć od warunków początkowych $x_1=0$, $x_2=10$, $x_3=27$. Dlatego wartość napięć początkowych na kondensatorach zadeklarowano następująco:

- ☐ Napięcie początkowe na kondensatorze C1: $V(1)_{t=0}=0[V]$.
- ☐ Napięcie początkowe na kondensatorze C2: $V(2)_{t=0}=0[V]$.




Rys.61. Przebieg zmiennej x_1 (opis w tekście).

□ Napięcie początkowe na kondensatorze C3: $V(3)_{t=0}=27[V]$.

W instrukcji .TRAN użyto słowa kluczowego UIC. Wartości początkowe potencjałów węzłowych nie będą zatem obliczane. Zostaną przyjęte te wartości, które określono w deklaracjach kondensatorów.

Wyniki analizy mogą być zaskakujące. Przebieg zmiennej x_1 (potencjał $V(1)$) pokazano na Rys. 61. Przebieg wygląda tak jak gdyby na sygnał sinusoidalny nałożony był szum o charakterze wybuchowym. Obserwujemy deterministyczny chaos. Podobny przebieg mają zmienne x_2 i x_3 (potencjały węzłowe $V(2)$ i $V(3)$).

Trajektoria fazowa układu równań (65), reprezentująca zależność między zmiennymi x_1 , x_2 , x_3 jest krzywą w przestrzeni trójwymiarowej. Na płaszczyźnie możliwe jest jedynie narysowanie jej wybranego rzutu. Niech płaszczyzną, na którą rzutujemy, będzie płaszczyzna określona równaniem: $x_3=0$. Wynik rzutowania trajektorii fazowej przedstawia Rys. 62. Na osi poziomej odłożona jest zmienna x_2 (potencjał $V(2)$) natomiast na osi pionowej odłożona jest zmienna x_1 (potencjał $V(1)$). Sama krzywa jest dość regularną ósemką ale poszczególne jej „zwoje” nawinięte są zupełnie nieregularnie. Używając programu PROBE można wykreślić rzut trajektorii na dowolną płaszczyznę. Za każdym razem przekonujemy się w jak nieregularny sposób tworzona jest krzywa. Co więcej, niezależnie od warunków początkowych



Rys.62. Rzut trajektorii fazowej na płaszczyznę $x_3=0$.

rozwiązanie układu równań (65) szybko zostaje ściągnięte do krzywej przedstawionej na Rys. 62. Krzywą o tej własności nazywa się atraktorem. Ten, który przedstawiony jest na Rys. 62 nazywany jest **atraktorem Lorentza**. Dokładnie rzecz biorąc nie udowodniono jeszcze, że przedstawiona krzywa jest atraktorem. Wszystko co wiadomo na temat tej krzywej uzyskano w wyniku symulacji komputerowej.

4.4.2. Cząstka amoniaku [8]

Inny przykład zastosowania programu PSpice do rozwiązywania zagadnień nieelektrycznych dotyczy mechaniki kwantowej. Rozważmy cząstkę amoniaku o wzorze chemicznym NH_3 . Schemat budowy przestrzennej tej cząstki przedstawia Rys. 63. Jak widać atom azotu może znajdować się ponad płaszczyzną wyznaczoną przez atomy wodoru w stanie $|1\rangle$ lub pod tą płaszczyzną w stanie $|2\rangle$ ⁶. Ogólnie rzecz biorąc nigdy nie wiadomo z całą pewnością, gdzie atom azotu się znajduje. W związku z tym stan cząstki amoniaku


⁶Oznaczenia stanów stosowane w mechanice kwantowej.

znalezienia cząstki w stanie $|1\rangle$ i w stanie $|2\rangle^7$. Liczby $C1$ i $C2$ są liczbami zespolonymi. Prawdopodobieństwo $P1$ znalezienia cząstki w stanie $|1\rangle$ wyraża się wzorem:

$$P1 = C1 \cdot \overline{C1} \quad (66)$$

Podobnie prawdopodobieństwo $P2$ znalezienia cząstki w stanie $|2\rangle$ wyraża się wzorem:

$$P2 = C2 \cdot \overline{C2} \quad (67)$$



Jest rzeczą oczywistą, że jeżeli jedynymi dopuszczalnymi przez nas stanami cząstki są wymienione dwa stany to prawdopodobieństwa $P1$ i $P2$ muszą spełniać związek:

$$P1 + P2 = 1 \quad (68)$$

Ewolucję czasową amplitud prawdopodobieństwa $C1$ i $C2$ opisuje równanie Hamiltona.

$$\begin{cases} i\hbar \frac{dC1}{dt} = H_{11} \cdot C1 + H_{12} \cdot C2 \\ i\hbar \frac{dC2}{dt} = H_{21} \cdot C1 + H_{22} \cdot C2 \end{cases} \quad (69)$$

Litera i oznacza tutaj czynnik urojony tzn. $i^2 = -1$, natomiast $\hbar = 1,055 \cdot 10^{-34}$ [J·s] to stała Plancka. Współczynniki H_{11} i H_{22} to odpowiednio energia cząstki w stanie $|1\rangle$ i energia cząstki w stanie $|2\rangle$. Współczynniki H_{12} i H_{21} opisują „przeskakiwanie” cząstki między wymienionymi stanami i spełniają następujący związek:

$$H_{12} = \overline{H_{21}} \quad (70)$$

Jeżeli brak jest pola elektrycznego to energia cząstki w obu stanach jest identyczna, a zatem $H_{11} = H_{22} = E_0$. Ze względu na symetrię stanów $H_{12} = H_{21} = -A$. Równania Hamiltona przyjmują następującą postać:

Cząstka amoniaku jest złożonym układem kwantowym. Trudno jest zatem podać teoretyczną wartość energii E_0 oraz współczynnika A . Wartości te mogą być wyznaczone

⁷Nie interesuje nas ruch cząstki w przestrzeni ani jej wewnętrzne stany wzbudzone.

$$\begin{cases} i\hbar \frac{dC1}{dt} = E_0 \cdot C1 - A \cdot C2 \\ i\hbar \frac{dC2}{dt} = -A \cdot C1 + E_0 \cdot C2 \end{cases} \quad (71)$$

doświadczalnie. Rozwiązanie układu równań (71) jest takie, że amplitudy prawdopodobieństwa stanowią pewną liniową kombinację funkcji:

$$\exp\left(\frac{E_0 - A}{i\hbar} \cdot t\right) \quad \text{oraz} \quad \exp\left(\frac{E_0 + A}{i\hbar} \cdot t\right) \quad (72)$$

Oznacza to, że stan cząstki amoniaku może być opisywany na dwa *równoważne* sposoby:

- Przez podanie amplitudy prawdopodobieństwa $C1$ znalezienia cząstki w stanie $|1\rangle$ i amplitudy $C2$ znalezienia cząstki w stanie $|2\rangle$, tak jak opisano to na początku paragrafu.
- Przez podanie amplitudy prawdopodobieństwa CI znalezienia cząstki w stanie o energii $E_I = E_0 + A$ oraz amplitudy prawdopodobieństwa CII znalezienia cząstki w stanie o energii $E_{II} = E_0 - A$.

Istnienie dwóch stanów cząstki amoniaku $|1\rangle$ i $|2\rangle$, różniących się położeniem atomu azotu w stosunku do płaszczyzny wyznaczonej przez atomy wodoru, jest równoważna istnieniu dwóch bliskich sobie stanów energetycznych. Przejście cząstki od stanu o energii E_I do stanu o niższej energii E_{II} wiąże się z wyemitowaniem kwantu promieniowania (fotonu) o częstotliwości $f_0 = (E_I - E_{II}) / (2\pi\hbar) = 2 \cdot A / (2\pi\hbar) \approx 24[\text{GHz}]$. Częstotliwość ta, w widmie promieniowania elektromagnetycznego, leży w obszarze mikrofal.

Rozważmy teraz w jaki sposób cząstka amoniaku oddziaływać będzie z polem elektrycznym zmieniającym się z częstotliwością równą f_0 . Tak jak pokazano to na Rys. 63 cząstka posiada moment elektryczny \underline{m} skierowany od atomu azotu w stronę płaszczyzny tworzonej przez atomy wodoru. Energia cząstki w obecności pola elektrycznego w stanie $|1\rangle$ różni się od energii w stanie $|2\rangle$. Załóżmy, że na Rys. 63 wektor natężenia pola elektrycznego $\underline{\varepsilon}(t)$ ma kierunek pionowy i jest skierowane z dołu do góry. Współczynnik H_{11} równania Hamiltona oznaczający energię cząstki w stanie $|1\rangle$ przyjmuje postać:

$$H_{11} = E_0 + m \cdot \varepsilon(t) \quad (73)$$

Współczynnik H_{22} oznaczający energię cząstki w stanie $|2\rangle$ ma postać:

$$H_{22} = E_0 - m \cdot \varepsilon(t) \quad (74)$$

Pozostałe dwa współczynniki pozostają niezmiennicze. Równania Hamiltona opisujące ewolucję czasową amplitud prawdopodobieństwa znalezienia układu w stanie $|1\rangle$ i w stanie $|2\rangle$ przyjmują postać:

$$\begin{cases} i\hbar \frac{dC1}{dt} = [E_0 + m \cdot \varepsilon(t)] \cdot C1 - A \cdot C2 \\ i\hbar \frac{dC2}{dt} = -A \cdot C1 + [E_0 - m \cdot \varepsilon(t)] \cdot C2 \end{cases} \quad (75)$$

Dokonajmy teraz podstawienia:

$$CI = \frac{[C1 - C2]}{\sqrt{2}} ; \quad CII = \frac{[C1 + C2]}{\sqrt{2}} \quad (76)$$

Funkcje CI i CII to wspomniane wcześniej amplitudy prawdopodobieństwa znalezienia układu w stanie o energii E_I i E_{II} . Równania Hamiltona mają teraz postać:

$$\begin{cases} i\hbar \frac{dCI}{dt} = E_I \cdot CI + m \cdot \varepsilon(t) \cdot CII \\ i\hbar \frac{dCII}{dt} = m \cdot \varepsilon(t) \cdot CI + E_{II} \cdot CII \end{cases} \quad (77)$$

Załóżmy, że rozwiązanie naszych równań można przedstawić następująco:

$$CI(t) = DI(t) \cdot \exp\left(\frac{E_I \cdot t}{i\hbar}\right) ; \quad CII(t) = DII(t) \cdot \exp\left(\frac{E_{II} \cdot t}{i\hbar}\right) \quad (78)$$

Po przekształceniach i podstawieniu $\omega_0 = 2\pi f_0 = (2A)/\hbar$ można zapisać w formie:

$$\begin{cases} i\hbar \frac{dDI(t)}{dt} = m \cdot \varepsilon(t) \cdot \exp(-i\omega_0 t) \cdot DII(t) \\ i\hbar \frac{dDII(t)}{dt} = m \cdot \varepsilon(t) \cdot \exp(i\omega_0 t) \cdot DI(t) \end{cases} \quad (79)$$

Zauważmy, że prawdopodobieństwo PI znalezienia cząstki w stanie o energii E_I oraz prawdopodobieństwo PII znalezienia cząstki w stanie E_{II} wyrażają się wzorami:

$$\begin{aligned} PI &= CI \cdot \overline{CI} = DI \cdot \overline{DI} \\ PII &= CII \cdot \overline{CII} = DII \cdot \overline{DII} \end{aligned} \quad (80)$$

Założmy, że pole elektryczne zmienia się sinusoidalnie z pulsacją ω :

$$\varepsilon(t) = \varepsilon_0 \cdot \sin(\omega t) \quad (81)$$

Niech cząstka NH_3 w chwili $t=0$ znajduje się w stanie o energii E_I (wyższej). Oznacza to, że $DI(0)=1$ oraz $DII(0)=0$. Równania Hamiltona mają postać:

$$\begin{cases} i\hbar \frac{dDI(t)}{dt} = m \cdot \varepsilon_0 \cdot \sin(\omega t) \cdot \exp(-i\omega_0 t) \cdot DII(t) \\ i\hbar \frac{dDII(t)}{dt} = m \cdot \varepsilon_0 \cdot \sin(\omega t) \cdot \exp(-i\omega_0 t) \cdot DI(t) \end{cases} \quad (82)$$

Równania te będziemy chcieli rozwiązać za pomocą programu PSpice w dwóch przypadkach:

- ☐ W przypadku rezonansu, tzn. gdy częstotliwość pola elektrycznego $\omega/(2\pi)$ jest identyczna z częstotliwością f_0 kwantu promieniowania emitowanego przez cząstkę przy przejściu od stanu o wyższej energii E_I do stanu o niższej energii E_{II} .
- ☐ W przypadku gdy częstotliwość pola elektrycznego będzie się nieco różniła od częstotliwości rezonansowej f_0 .

Aby uzyskać uniwersalne rozwiązania równań (82) będziemy się posługiwali znormalizowanym czasem τ oraz znormalizowanymi pulsacjami ω' , ω'_0 określonymi zgodnie ze wzorami:

$$\begin{aligned} \Omega &= \frac{m \cdot \varepsilon_0}{\hbar} ; \quad \tau = \Omega \cdot t \\ \omega' &= \frac{\omega}{\Omega} ; \quad \omega'_0 = \frac{\omega_0}{\Omega} \end{aligned} \quad (83)$$

Równania przyjmują teraz postać:

$$\begin{cases} i \frac{dDI(\tau)}{d\tau} = \sin(\omega' \tau) \cdot \exp(-i\omega'_0 \tau) \cdot DII(\tau) \\ i \frac{dDII(\tau)}{d\tau} = \sin(\omega' \tau) \cdot \exp(-i\omega'_0 \tau) \cdot DI(\tau) \end{cases} \quad (84)$$

Układ dwóch równań różniczkowych (84) o współczynnikach zespolonych można rozwiązywać za pomocą programu PSpice po uprzednim rozbiciu go na układ czterech równań (85), w których występują już tylko funkcje rzeczywiste tzn. część rzeczywista i urojona funkcji $DI(\tau)$ i $DII(\tau)$.

$$\left\{ \begin{array}{l} \frac{d \operatorname{im}(DI)}{d\tau} = -\sin(\omega'\tau)\cos(\omega_0'\tau) \cdot \operatorname{re}(DII) + \sin(\omega'\tau)\sin(\omega_0'\tau) \cdot \operatorname{im}(DII) \\ \frac{d \operatorname{re}(DI)}{d\tau} = \sin(\omega'\tau)\cos(\omega_0'\tau) \cdot \operatorname{im}(DII) + \sin(\omega'\tau)\sin(\omega_0'\tau) \cdot \operatorname{re}(DII) \\ \frac{d \operatorname{im}(DII)}{d\tau} = -\sin(\omega'\tau)\cos(\omega_0'\tau) \cdot \operatorname{re}(DI) - \sin(\omega'\tau)\sin(\omega_0'\tau) \cdot \operatorname{im}(DI) \\ \frac{d \operatorname{re}(DII)}{d\tau} = \sin(\omega'\tau)\cos(\omega_0'\tau) \cdot \operatorname{im}(DI) - \sin(\omega'\tau)\sin(\omega_0'\tau) \cdot \operatorname{re}(DI) \end{array} \right. \quad (85)$$

Dla równań (85) można zbudować analog elektryczny tzn. obwód elektryczny opisywany tymi właśnie równaniami. Analog ten przedstawiony jest na Rys. 64. Występują tam cztery sekcje, każda z jednym kondensatorem o pojemności 1[F]. Kolejne sekcje odpowiadają kolejnym równaniom w układzie (85). Potencjały V(1), V(2), V(3) i V(4) odpowiadają kolejno części urojonej DI, części rzeczywistej DI, części urojonej DII i części rzeczywistej DII. Pozostałe sekcje służą do uzyskania :

- ☐ Przebiegu sinusoidalnego o pulsacji ω' — V(10).
- ☐ Przebiegu sinusoidalnego o pulsacji ω_0' — V(20).
- ☐ Przebiegu kosinusoidalnego o pulsacji ω_0' — V(30).

Zadeklarowanie wymuszenia w postaci kosinusoidy stanowi w programie PSpice pewien problem. W przebiegu typu SIN brak jest parametru, który pozwalałby zmienić fazę początkową. Wobec tego użyto źródła o przebiegu sinusoidalnym i częstotliwości dwa razy mniejszej niż wymagana. Kosinusoidę można uzyskać z tego przebiegu na podstawie wzoru:

$$\cos(\omega t) = 1 - 2 \cdot \sin^2(\omega t) \quad (86)$$

Niezbędne jest użycie nieliniowego źródła sterowanego. W przypadku naszego obwodu przebieg kosinusoidalny ma potencjał węzła V(30).

Z punktu widzenia programu PSpice obwód przedstawiony na Rys. 64 jest patologiczny. Węzły 1, 2, 3 i 4 nie mają stałoprądowego połączenia z węzłem masy. Aby temu zaradzić każdy z wymienionych węzłów połączono z węzłem masy opornikiem o wartości 10[MΩ]. Opornik taki będzie miał niezauważalny wpływ na rozwiązanie równań obwodu, natomiast z punktu widzenia programu PSpice struktura obwodu stanie się poprawna. Dane do obliczeń dla programu PSpice przedstawione są poniżej.

RÓWNANIA CZĄSTKI AMONIAKU - REZONANS
.OPTIONS ITL5=10000

```

CI1 1 0 1
RI1 1 0 10MEG
GI11 0 1 POLY(3) 10 0 30 0 4 0 0 0 0 0 0 0 0 0 0 0 -1
GI12 0 1 POLY(3) 10 0 20 0 3 0 0 0 0 0 0 0 0 0 0 0 1

CR1 2 0 1 IC=1
RR1 2 0 10MEG
GR11 0 2 POLY(3) 10 0 30 0 3 0 0 0 0 0 0 0 0 0 0 0 1

```

```

GR12 0 2 POLY(3) 10 0 20 0 4 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1
CI2 3 0 1
RI2 3 0 10MEG
GI21 0 3 POLY(3) 10 0 30 0 2 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1
GI22 0 3 POLY(3) 10 0 20 0 1 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1

CR2 4 0 1
RR2 4 0 10MEG
GR21 0 4 POLY(3) 10 0 30 0 1 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1
GR22 0 4 POLY(3) 10 0 20 0 2 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1

R10 10 0 1
V10 10 0 SIN(0 1 10)

R20 20 0 1
V20 20 0 SIN(0 1 10)

R40 40 0 1
V40 40 0 SIN(0 1 5)
R30 30 0 1
E30 30 0 POLY(1) 40 0 1 0 -2

.TRAN 0.1 10 UIC
.PROBE V(1) V(2) V(3) V(4)
.END
RÓWNANIA CZĄSTKI AMONIAKU - POZA REZONANSEM
.OPTIONS ITL5=10000

CI1 1 0 1
RI1 1 0 10MEG
GI11 0 1 POLY(3) 10 0 30 0 4 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1
GI12 0 1 POLY(3) 10 0 20 0 3 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1

CR1 2 0 1 IC=1
RR1 2 0 10MEG
GR11 0 2 POLY(3) 10 0 30 0 3 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1
GR12 0 2 POLY(3) 10 0 20 0 4 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1


CI2 3 0 1
RI2 3 0 10MEG
GI21 0 3 POLY(3) 10 0 30 0 2 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1
GI22 0 3 POLY(3) 10 0 20 0 1 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1

CR2 4 0 1
RR2 4 0 10MEG
GR21 0 4 POLY(3) 10 0 30 0 1 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1
GR22 0 4 POLY(3) 10 0 20 0 2 0      0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 -1

R10 10 0 1
V10 10 0 SIN(0 1 10.5)
R20 20 0 1
V20 20 0 SIN(0 1 10)
R40 40 0 1
V40 40 0 SIN(0 1 5)
R30 30 0 1
E30 30 0 POLY(1) 40 0 1 0 -2
.TRAN 0.1 10 UIC
.PROBE V(1) V(2) V(3) V(4)
.END

```


Wartość momentu elektrycznego cząstki amoniaku wynosi 1,46[Debye] [27] co odpowiada $4,87 \cdot 10^{-30}$ [A·s·m]. Jeżeli założymy amplitudę pola elektrycznego równą 100 [V/cm] to wartość znormalizowanej pulsacji rezonansowej powinna wynosić 327,0. Nasze obliczenia wykonywane będą dla znormalizowanej pulsacji rezonansowej równej 10,0 tzn. zakładamy, że pole elektryczne jest prawie 33 razy większe. W przypadku obliczeń poza rezonansem założono, że częstotliwość rezonansowa różni się od częstotliwości pola elektrycznego o 5,0[%].



Rys.64. Analog elektryczny równań opisujących cząstkę amoniaku.

Wynik obliczeń przedstawiony jest na Rys. 65. Na osi poziomej odłożony jest znormalizowany czas τ . Na osi pionowej odłożone jest prawdopodobieństwo znalezienia cząstki w stanie o wyższej energii E_I (wzór (80)). W przypadku , gdy pole elektryczne zmienia się z częstotliwością równą częstotliwości rezonansowej obserwujemy emisję wymuszoną. Pole elektryczne powoduje, że cząstka cyklicznie przechodzi od stanu o energii E_I do stanu o energii E_{II} . Wiąże się to z wypromieniowywaniem i pochłanianiem odpowiedniego kwantu energii. Odchylenie częstotliwości pola od częstotliwości rezonansowej o 5,0% znacznie zmniejsza ten efekt. Nie ma już chwil, w których cząstka znajduje się w stanie o energii E_{II} z prawdopodobieństwem równym 1. Obserwowane na Rys. 65 zjawisko emisji wymuszonej dla cząstki NH_3 leży u podstaw działania masera amoniakalnego — generatora spójnego promieniowania mikrofaleowego o częstotliwości 24[GHz]. Zjawisko emisji wymuszonej występuje we wszystkich układach kwantowych, w których może dojść do spontanicznej emisji fotonu. W szczególności układy atomowe promieniujące w zakresie promieniowania widzialnego wykorzystywane są do budowy generatorów światła spójnego: laserów⁸.

⁸ang. Light (Microwave) Amplification by Stimulated Emission of Radiation — wzmocnienie światła (mikrofal) przez wymuszoną emisję promieniowania.



Rys.65. Prawdopodobieństwo znalezienia cząstki amoniaku w wyższym stanie energetycznym; (biały kwadrat) rezonans; (czarny kwadrat) poza rezonansem.